/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */
configurations.register('documentation') {
    canBeConsumed = false
    canBeResolved = true
    attributes {
        attribute(Category.CATEGORY_ATTRIBUTE, objects.named(Category, Category.LIBRARY))
        attribute(Bundling.BUNDLING_ATTRIBUTE, objects.named(Bundling, Bundling.EXTERNAL))
        attribute(Usage.USAGE_ATTRIBUTE, objects.named(Usage, Usage.JAVA_RUNTIME))
    }
}

dependencies {
    add('documentation', platform(project(':grails-bom')))
    add('documentation', 'org.fusesource.jansi:jansi')
    add('documentation', 'jline:jline')
    add('documentation', 'com.github.javaparser:javaparser-core')
    add('documentation', 'org.apache.groovy:groovy')
    add('documentation', 'org.apache.groovy:groovy-groovydoc')
    add('documentation', 'org.apache.groovy:groovy-ant')
    add('documentation', 'org.apache.groovy:groovy-docgenerator')
    add('documentation', 'org.apache.groovy:groovy-templates')
}

String resolveProjectVersion(String artifact) {
    String version = configurations.runtimeClasspath
            .resolvedConfiguration
            .resolvedArtifacts
            .find {
                it.moduleVersion.id.name == artifact
            }?.moduleVersion?.id?.version
    if (!version) {
        return null
    }
}

tasks.withType(Groovydoc).configureEach { Groovydoc gdoc ->
    gdoc.exclude('META-INF/**', '*yml', '*properties', '*xml', '**/Application.groovy', '**/Bootstrap.groovy', '**/resources.groovy')
    gdoc.groovyClasspath = configurations.documentation
    gdoc.windowTitle = "${project.findProperty('pomArtifactId') ?: project.name} - $projectVersion"
    gdoc.docTitle = "${project.findProperty('pomArtifactId') ?: project.name} - $projectVersion"
    gdoc.access = GroovydocAccess.PROTECTED
    gdoc.includeAuthor = false
    gdoc.includeMainForScripts = false
    gdoc.processScripts = false
    gdoc.noTimestamp = true
    gdoc.noVersionStamp = false

    doFirst {
        def gebVersion = resolveProjectVersion('geb-spock')
        if(gebVersion) {
            gdoc.link("https://groovy.apache.org/geb/manual/${gebVersion}/api/", 'geb.')
        }

        def testContainersVersion = resolveProjectVersion('testcontainers')
        if(testContainersVersion) {
            gdoc.link("https://javadoc.io/doc/org.testcontainers/testcontainers/${testContainersVersion}/", 'org.testcontainers.')
        }

        def springVersion = resolveProjectVersion('spring-core')
        if(springVersion) {
            gdoc.link("https://docs.spring.io/spring-framework/docs/${springVersion}/javadoc-api/", 'org.springframework.core.')
        }

        def springBootVersion = resolveProjectVersion('spring-boot')
        if(springBootVersion) {
            gdoc.link("https://docs.spring.io/spring-boot/docs/${springBootVersion}/api/", 'org.springframework.boot.')
        }
    }
}